

## Plotting Data from a File with Gnuplot

This is a brief introduction by example to using the GNUPLOT plotting program to make XY plots of data from a file. We'll use command-line arguments only (but you may also have access to a graphical interface). Start up GNUPLOT by typing “gnuplot”, which should give you some information and then a prompt `gnuplot>` at which you can type commands.

You can get general help by typing `help` at the `gnuplot>` prompt or help on a specific command, e.g., `gnuplot> help plot` will tell you about the plot command.

Here is the file `quadratic_eq.dat`, which was generated by a test program exploring subtractive cancellations in formulas for the roots of a quadratic equation  $ax^2 + bx + c = 0$  in Session 2.

```
# Calculation of quadratic equation roots in single precision
#      1/c      |relative error 1|  |relative error 2|
1.000000e+01    4.355647e-07    4.893942e-07
1.000000e+02    1.765078e-06    1.852387e-06
1.000000e+03    3.503225e-05    3.505630e-05
1.000000e+04    1.410045e-04    1.409685e-04
1.000000e+05    1.355599e-03    1.353804e-03
1.000000e+06    4.632579e-02    4.857613e-02
1.000000e+07    1.920930e-01    1.611393e-01
# 1.000000e+08    1.000000e+00    inf
```

Shown are the values of  $1/c$  (where  $a = 1$  and  $b = 2$ ) and the relative error in the “bad” way of calculating the first and second roots (with respect to the “good” way). The `#`'s on any line tell GNUPLOT to ignore everything to the right of `#` (i.e., these are comments). The last line is commented out to avoid problems with the “inf” result.

We can make a quick plot of the 2nd column versus the first with:

```
gnuplot> plot "quadratic_eq.dat" using 1:2
```

then switch to log scales and replot:

```
gnuplot> set logscale # sets BOTH axes to log scales
```

```
gnuplot> replot
```

then plot the 3rd column AND the 2nd column versus the 1st:

```
gnuplot> plot "quadratic_eq.dat" using 1:2, "quadratic_eq.dat" using 1:3
```

That's all there is!

You can set the x and y ranges and add titles and labels and move the key and so on. Here is a sample transcript and the graph it generates (use `help` for more information on any command).

```
gnuplot> set title 'Landau/Paez 3.4-1 Quadratic Equation'
gnuplot> set xlabel 'c=10^{-n} for n=1..7 [a=1,b=2]'
gnuplot> set ylabel 'relative error'
gnuplot> set logscale
gnuplot> set xrange [1:1e7]
gnuplot> set pointsize 1.5 # set the size of the plotted points
gnuplot> set key top left # move the key away from the lines
gnuplot> set timestamp # turn on a date/time indicator
gnuplot> plot "quadratic_eq.dat" using 1:2 title '1st root',\
> "quadratic_eq.dat" using 1:3 title '2nd root'
gnuplot>
gnuplot> set out "quadratic_eq.ps" # an output postscript file
gnuplot> set terminal postscript # switch to postscript mode
Terminal type set to 'postscript'
Options are 'landscape noenhanced monochrome dashed defaultplex "Helvetica" 14'
gnuplot> replot # plot to the file
gnuplot> quit
```

